第一章 Delphi 與 Resource File
Resource是指可以嵌入Win32程式的二進位資料, 它可以是Windows標準的resource, 也可以是使用者自訂(application-defined)的. 標準的resource像是Icon, Cursor, Bitmap, Font, String或Version Information等. 

問題1-1: 如何產生 .res 檔

產生 .res檔的方法有下列幾種:

1.使用resource script(.rc檔)與BRCC32.Exe

2.使用Borland Resource Workshop, Nowadays這類的resource editor.

3.直接依resource的格式將其resources寫入檔案內

這裏我們介紹第一種方法.

所謂resource script(.rc檔)是文字檔, 其內容列出有哪些resources, 它的格式如下:

ResName1  ResType1  ResFileName1
ResName2  ResType2  ResFileName2 
... 
ResName欄位指定一個識別該行resource的唯一名稱或整數值;  ResType欄位指該行resource的型態, resource型態列舉如下:

	RT_ACCELERATOR
	Accelerator table

	RT_BITMAP
	Bitmap resource

	RT_DIALOG
	Dialog box

	RT_FONT
	Font resource

	RT_FONTDIR
	Font directory resource

	RT_MENU
	Menu resource

	RT_RCDATA
	Application-defined resource(raw data)

	RT_STRING
	String-table entry

	RT_MESSAGETABLE
	Message-table entry

	RT_CURSOR
	Hardware-dependent cursor resource

	RT_GROUP_CURSOR
	Hardware-independent cursor resource

	RT_ICON
	Hardware-dependend icon resource

	RT_GROUP_ICON
	Hardware-independent icon resource

	RT_VERSION
	Version resource


ResFileName欄位則指該resource的檔案名稱, 若該檔案與 .rc檔不在同一路徑的話, 則須加上檔案的完整路徑.

首先產生一個文字檔, 將其更名為sample.rc, 記住 .rc的主檔名不可與Delphi的 任一個unit檔與project檔相同名稱(因為Delphi會為Project及每個unit產生 .res檔). 編輯完成後, 假設存在C:\Test目錄。以下是sample.rc檔的內容:

1  BITMAP  "C:\Test\1.bmp"
MyBmp  BITMAP  "C:\WINNT\Winnt256.bmp"
MyJpeg  JPEG  beauty.jpg

1  ICON  dialog.ico

car  ICON  car.ico

pikacur  CURSOR  pikacur.cur

dunk  AVI  dunk.avi
start  WAVE  "C:\WINNT\Media\start.wav"
Lsansuni  MYFONT  "C:\Windows\Font\Lsansuni.ttf"
TestHtml  23  test1.html

HtmlGif  2110  4in1.gif

接下來進入 MS-Dos 模式, 準備Compile剛完成的sample.rc檔. 假設Delphi裝在C:\Program Files目錄, 執行Command如下:

C:\Test>C:\Program Files\Borland\Delphi5\Bin\BRCC32 sample.rc

執行完畢, 在C:\Test的目錄下即產生一個sample.RES檔。

問題1-2: 如何在 Delphi 內連結 .res檔

在Delphi專案中的Main Form的Unit找到 {$R *.DFM} 這行, 在這行的下方加上 {$R sample.res} 或 {$RESOURCE sample.res}, 這樣即可將sample.res這個resource file載入Delphi的專案中。

$R是Compiler Directives之一, 其格式為 {$R 檔案名稱} 或 {$Resource 檔案名稱}。這個編譯指令的用意在指定要含入指定的資源檔. 這資源檔必須是標準的Windows resource file, 且副檔名是 .RES. 如果檔案名稱中有空白的話, 須用單引號, 例: {$R ‘My resource’}.

問題1-3: 如何在 Delphi 內得到一特定modules內所有resource的names, types的列表

使用API: LoadLibraryEx, FreeLibrary與EnumResourceNames

var

  LastNode: TTreeNode;

  LastType: string;

const

  RESSTRINGS: array[1..22] of string = ('RT_CURSOR', 'RT_BITMAP', 'RT_ICON', 'RT_MENU', 'RT_DIALOG', 'RT_STRING', 'RT_FONTDIR', 'RT_FONT', 'RT_ACCELERATOR', 'RT_RCDATA', 'RT_MESSAGETABLE',                'RT_GROUP_CURSOR', 'RT_13', 'RT_GROUP_ICON', 'RT_15',                'RT_VERSION', 'RT_DLGINCLUDE', 'RT_18', 'RT_PLUGPLAY',                'RT_VXD', 'RT_ANICURSOR', 'RT_ANIICON');

// EnumResNameProc為EnumResourceNames所須的Callback function

function EnumResNameProc(hModule: THandle; lpszType, lpszName: PChar;

  lParam: LongInt): Boolean; stdcall;

var

  sType: string;

  tvItem: TTreeNode;

begin

  if (longint(lpszType) > 256) then

    sType := string(lpszType)

  else

    sType := RESSTRINGS[LongInt(lpszType)];

  if sType <> LastType then

  begin

    LastNode := Form1.TreeView1.Items.Add(nil, sType);

    LastType := sType;

  end;

  if LongInt(lpszName) < 65535 then

    tvItem := Form1.TreeView1.Items.AddChild(LastNode, IntToStr(LongInt(lpszName)))

  else

    tvItem := Form1.TreeView1.Items.AddChild(LastNode, string(lpszName));

  end;

  Result := True;

end;

procedure TForm1.btnEnumResourceNamesClick(Sender: TObject);

var

  hMod: Thandle;

  resType: Byte;

begin

  LastType := '';

  TreeView1.Items.Clear;

  hMod := LoadLibraryEx(ParamStr(0)), 0, LOAD_LIBRARY_AS_DATAFILE);

  if hMod = 0 then

    exit

  else

  try

    for resType := 1 to 21 do

      EnumResourceNames(hMod, MakeIntResource(resType), @EnumResNameProc, 0);

  finally

    FreeLibrary(hMod);

  end;

end;
問題1-4: 讀出resource的相關API
1.以LoadLibrary/LoadLirbaryEx將一個EXE或DLL檔載入記憶體, 如果是使用本身EXE的resource則毋須此步驟.

2.以FindResource/FindResourceEx取得module內resource的handle.

3.LoadResource利用2.步驟取得的handle, 將指定的resource載入記憶體, 後以FreeResource來釋放以LoadResource取得的resource

4.以3.步驟取得的Handle為LockResource的參數, 取得指定resource的指標.

5.若有使用LoadLibrary/LoadLibraryEx, 則須使用FreeLibrary來釋放.
接下來便以如何產生 .res檔一節說明方式所產生的sample.res來講解接下來的問題。

問題1-5: 如何自resources中讀入Bitmap

這裏介紹TBitMap的LoadFromResourceName與LoadFromResourceID的方法.

1.使用TBitmap類別的LoadFromResourceName method

procedure TForm1.Button1Click(Sender: TObject);

begin
  // samplre.rc中MyBmp BITMAP “C:\WINNT\Winnt256.bmp”這行

// ResName為MyBmp

  Image1.Picture.Bitmap.LoadFromResourceName(hInstance, 'MyBmp');

end;
2.使用TBitmap類別的LoadFromResourceID method

var

 bmp: TBitmap;

begin

  bmp := TBitmap.Create;
  try

  // LoadFromResourceID中的1這個參數

// 即指sample.rc檔中1 BITMAP “C:\Test\1.bmp”這行

// 該行的ResName為1

    BbmpLoadFromResourceID(hInstance, 1);

    Image1.Canvas.Draw(0, 0, bmp);
  finally

    bmp.Free;
  end;

end;

問題1-6: 如何自resources中讀入Jpeg

底下的範例將resource中的ResName為Kaoru的Jpeg畫在Form1的Canvas上

uses Jpeg;

procedure TForm1.btnJpegFromResourceClick(Sender: TObject);
var

  hRes: THandle;

  hMem: THandle;

  MemStream: TMemoryStream;

  pRes: PByte;

  ResSize: Longint;

  JPEGImage: TJPEGImage;

begin

  MemStream := TMemoryStream.Create;

  JPEGImage := TJPEGImage.Create;
Try

  // 取得ResName為Kaoru在hInstace中的Handle

hRes := FindResource(hInstance, PChar('Kaoru'), 'JPEG');
// 將hRes這個handle所指的resource載入memory

hMem := LoadResource(hInstance, hRes);
// 將hMem在memory中所指的resource lock住

// 若mark掉LockResource這行, 會產生 ”Jpeg Error #52” 的錯誤

  pRes := LockResource(hMem);
  // SizeOfResource取得指定resource的Byte數

    ResSize := SizeOfResource(hInstance, hRes);
    // 設定 memory stream的size property

  MemStream.SetSize(ResSize);

    MemStream.Write(pRes^, ResSize);

  FreeResource(hMem);
    // 將指標移回0的位置, 否則會產生 ”Jpeg Error #41” 的錯誤

    MemStream.Seek(0, 0);
JPEGImage.LoadFromStream(MemStream);

    Canvas.Draw(0, 0, JPEGImage);
  finally

  JPEGImage.Free;

    MemStream.Free;
  End;

End;

問題1-7: 如何自resources中讀入Icon
1.用LoadIcon的方式

procedure TForm1.btnLoadIconFromResClick(Sender: TObject);
begin
  Self.Icon.Handle := LoadIcon(hInstance, MAKEINTRESOURCE(1));

  Application.Icon.Handle := LoadIcon(hInstance, PChar('car'));

end;
2.用ExtractIcon的方式

procedure TForm1.btnExtractIconClick(Sender: TObject);

var 
i, iIconCount: integer;

  MyIcons: array of TIcon;

Begin
// 最後一個參數代入 -1 取得 project1.exe 內的Icon總數目

  iIconCount := ExtractIcon(hInstance, PChar(ParamStr(0)), Cardinal(-1));
SetLength(MyIcons, iIconCount); // 設定MyIcons陣列長度

  For i := 0 to iIconCount - 1 do
begin
    MyIcons[i] := TIcon.Create;

    Myicons[i].Handle := Extracticon(hInstance, PChar(ParamStr(0)), i);
    // 將取得的新Icon加入ImageList1元件

    ImageList1.AddIcon(Myicons[i]); 
    ComboBox1.Items.Add(IntToStr(i));

  end;

  combobox1.ItemIndex := 0;
  // 將Application的Icon替換成Index為0的這個Icon

  ImageList1.GetIcon(0, Application.Icon);

end;
3.用TResourceStream的方式

procedure TForm1.btnLoadCursorFromResStreamClick(Sender: TObject);
var 
ResStream1, ResStream2: TResourceStream;

Begin
  // TresourceStream.Create是以ResName為識別來取得所要之resource

  ResStream1 := TResourceStream.Create(hInstance, 'car', RT_ICON);

  try

    Application.Icon.LoadFromStream(ResStream1);

  finally

    ResStream1.Free;

  end;

  // TresourceStream.CreateFromID是以ResID為識別來取得所要之resource

  ResStream2 := TResourceStream.CreateFromID(hInstance, 1, RT_ICON);

  try

    Self.Icon.LoadFromStream(ResStream2);

  finally

    ResStream2.Free;

  end;

end;
問題1-8: 如何自resources使用Cursor

以LoadCursor的方式取得resource內ResName為pikacur的Cusor

procedure TForm1.btnLoadCursorFromResClick(Sender: TObject);

const MyCursor = 1;

begin

  Screen.Cursors[MyCursor] := LoadCursor(hInstance, PChar('pikacur'));

  Cursor := MyCursor; // 並將Form1的Cursor置換
end;
問題1-9: 如何自resources中播放Avi

procedure TForm1.btnPlayAviFromResClick(Sender: TObject);

begin

  Animate1.ResName := 'dunk';

  Animate1.ResHandle := hInstance;

  Animate1.Active := TRUE;

end;
注意: TAnimate元件只能播放沒有聲音的 .avi檔, 否則會出現 “Cannot open AVI” 的錯誤訊息.

問題1-10: 如何自resources中播放Wave

1.先將ResName為login的resource讀至Memory後, 再以SndPlaySound函式播放

procedure TForm1.btnPlayWaveFromResClick(Sender: TObject);

var

  hFind: THandle;

hRes: THandle;

  pWave: PChar;

begin

  hFind := FindResource(HInstance, PChar('login'), PChar('WAVE'));

  if hFind <> 0 then begin

    hRes := LoadResource(HInstance, hFind);

    if hRes <> 0 then begin

      pWave := LockResource(hRes);

      if Assigned(pWave) then

        SndPlaySound(pWave, snd_ASync or snd_Memory);

      UnlockResource(hRes);

    end;

    FreeResource(hFind);

  end;

end;
2.直接以PlaySound函式播放resource中ResName為login的wav

procedure TForm1.btnPlayWaveFromResClick(Sender: TObject);

begin

  PlaySound(Pchar(‘login’), hInstance, SND_RESOURCE or SND_SYNC);

end;
呼叫SndPlaySound與PlaySound函式, 須uses MMSystem這個unit.

問題1-11: 如何自resources中使用Font

procedure TForm1.btnUseFontFromResClick(Sender: TObject);

var

  ResStream: TResourceStream;

begin

  ResStream := TResourceStream.Create(hInstance, 'Lsansuni', PChar('MYFONT'));

  ResStream.SavetoFile('Lsansuni.ttf');

  ResStream.Free;
  // 呼叫AddFontResource將新字型加入windows font table中

  if AddFontResource(PChar('Lsansuni.ttf')) = 0 then begin

    ShowMessage(SysErrorMessage(GetLastError));

    Exit;

  end;
  // 通知其它window, window font table已有異動

  SendMessage(HWND_BROADCAST, WM_FONTCHANGE, 0, 0);

  btnUseFontFromRes.Font.Name := 'Lsansuni';

  btnUseFontFromRes.Font.Size := 12;

end;
問題1-12: 如何自resources中使用Html

sample.rc檔中的TestHtml 23 test1.html與HtmlGif 2110 4in1.gif這兩行中的23指ResType為RT_HTML, 2110指linked file.

Test1.html檔的內容如下:

<html>

<head>

<title>use HTML from resource in Delphi</title>

</head>

<body bgcolor=black>

  <p align=center>

  <b><font color=red size=5>using HTML from Delphi resource:</font></b>

  </p>
  <img src="HtmlGif">

</body>

</html>
其中HtmlGif為sample.rc檔中ResName為 HtmlGif的resource.
procedure TForm1.btnUseHtmlFromResClick(Sender: TObject);

var

  Flags, TargetFrameName, PostData, Headers: OleVariant;

  sURL: string;

begin

  sURL := 'res://' + Application.ExeName + '/TestHtml';
  // 將ResName為TestHtml的resource顯示在WebBrowser1上

  WebBrowser1.Navigate(sURL, Flags, TargetFrameName, PostData, Headers);

end;
res Protocol 用法為: res://sFile[/sType]/sID, 其中sFile指包含resource的完整檔名; sType指數字或字串的resource type; sID則指數字或字串的resource ID.

問題1-13: 如何自resources中使用String Table

String Table的 .rc檔內容與之前所述不同, 其格式如下:

STRINGTABLE

 {

  101, "This is string 1"

  102, "This is string 2"

  103, "This is string 3"

 }
或以下的格式也可以:

STRINGTABLE

BEGIN

  101, "This is string 1"

  102, "This is string 2"

  103, "This is string 3"
END

將其存檔為MyStringTable.rc, 利用BRCC32.EXE產生MyStringTable.res檔, 最後在程式中加上 {$R MyStringTable.RES} 將其連結至專案即可。使用String Table的範例如下:

1.使用LoadStr函式, 不過Delphi Help中說明, LoadStr函式是為了向之版本的Delphi相容而存在, 現可改用resourcestring的方式.

procedure TForm1.btnLoadStrClick(Sender: TObject);

begin

  ShowMessage(LoadStr(101)); 

  ShowMessage(LoadStr(102));

  ShowMessage(LoadStr(103));

end;
2.使用LoadString函式

procedure TForm1.btnLoadStringClick(Sender: TObject); 
var 
  buffer: array[0..255] of Char; 
begin 
  Loadstring(hinstance, 101, @buffer, 255); 
  ShowMessage(StrPas(buffer)); 
  Loadstring(hinstance, 102, @buffer, 255); 
  ShowMessage(StrPas(buffer)); 
  Loadstring(hinstance, 103, @buffer, 255); 
  ShowMessage(StrPas(buffer)); 
end;
問題1-14: 如何自resources中載入RTF至TrichEdit

1.使用TResourceStream的方式

procedure TForm1.btnLoadRTFFromResClick(Sender: TObject);

var ResStream: TResourceStream;

begin

  ResStream := TResourceStream.Create(hInstance, 'MyRtf', PChar('RTF'));

  try

    RichEdit1.Lines.LoadFromStream(ResStream);

  finally

    ResStream.Free;

  end;

end;
2.使用FindResource的方式

procedure TForm1.Button1Click(Sender: TObject);

const

   TmpFile = 'temp.rtf';

var

   hFind: THandle;

hRes: THandle;

   pData: PChar;

   sl: TStringList;

begin
  hFind := FindResource(hInstance, 'MyRtf', 'RTF');

if hFind = 0 then begin
ShowMessage(‘Could not find resource’);

Exit;

end;
hRes := LoadResource(hInstance, hFind);
if hRes <> 0 then begin

    pData := LockResource(hRes);

    sl := TStringList.Create;

    try

      sl.SetText(pData);

  sl.SaveToFile(TmpFile);

    finally

      sl.Free;

    UnlockResource(hRes);

      FreeResource(hRes);

    end ;

    RichEdit1.Lines.LoadFromFile(TmpFile);

    DeleteFile(TmpFile);

  end;
end;
問題1-15: 如何產生有VersionInfo的resource

VersionInfo的格式為: versionID  VERSIONINFO  fixed-info  { block-statement . . . }, 產生VersionInfo的 .rc檔內容範例如下:

1 VERSIONINFO

FILEVERSION 1, 0, 0, 0

PRODUCTVERSION 1, 0, 0, 0

FILEOS VOS__WINDOWS32

FILETYPE VFT__APP

{BLOCK "StringFileInfo" 
 {BLOCK "040904E4"
  {

   VALUE "CompanyName", "Delphi KTop\000"

   VALUE "FileDescription", "VersionInfo Resource Test\000"

   VALUE "FileVersion", "1, 0, 0, 1\000"

   VALUE "InternalName", "VersionInfo InternalName\000"

   VALUE "LegalCopyright", "Copyright KTop 2003\000"

   VALUE "OriginalFilename", "VersionInfo OriginalFileName\000"

   VALUE "ProductName", "KTop VersionInfo Test\000"

   VALUE "ProductVersion", "1, 0, 0, 1\000" 
  }

 }

}
編輯完成後將其存檔為VersionInfo.rc, 依之前所述的方式產生VersionInfo.res檔, 並連結至Delphi專案中.

VersionInfo.rc檔內容說明如下:

1 VERSIONINFO:這行的1指versionID, 它的值必須是1.

fixed info: 包括FileVersion, Product Version, FileOS, FileType等.

FILEVERSION 1, 0, 0, 0: 定義該file的二進位版本, 由4個16-bit的整數組成.

PRODUCTVERSION 1, 0, 0, 0: 定義product該product的二進位版本, 由4個16-bit的整數組成.

FILEOS VOS__WINDOWS32:指該file設計於何種OS平台, VOS__Windows32是指為32-bit windows所設計.

FILETYPE VFT__APP指該file的type: VFT_APP指該file是一個Application.

Block區塊: 指定一個以上的版本資訊區塊. 其格式如下:

BLOCK "StringFileInfo"  
{
 BLOCK "lang-charset"
 {          
VALUE "string" "string value"
}

}

0409950: 0409指langID, 其值為U.S. English; 950指charset, 其值為Taiwan(Big5).

"CompanyName": 指生產此file的Company, 此欄位為必須.

"FileDescription": 對該file的描述, 此欄位為必須.

"FileVersion": 該file的版本數字, 此欄位為必須

"InternalName": 如果file是一個DLL, InterName指一個module name; 若該file沒有internal name, 則此欄位值為原本的file name. 此欄位為必須.

"OriginalFileName": 該file的original name, 不包含路徑. 這項資訊可判斷使用者是否更改了file的name. 此欄位為必須.

"ProductName": 跟file一直分發出去的product的name. 此欄位為必須.

"ProductVersion": 上述product的版本. 此欄位為必須.
詳細資訊可參考MSDN中有關VersionInfo的說明.

問題1-16: 如何讀出resource中的VersionInfo
使用API: GetFileVersionInfoSize, GetFileVersionInfo, VerQueryValue.

procedure TForm1.btnReadVersionInfoFromResClick(Sender: TObject);

var

 VerSize: DWORD;

 Zero: THandle;

 pBlock: Pointer;
 sSubBlock: string;

 pBuffer: Pointer;

 Size: UINT;

begin

  // Get size of Version resource

  VerSize := GetFileVersionInfoSize(PChar(ParamStr(0)), Zero);

  If VerSize <> 0 Then

  Begin

    GetMem(pBlock, VerSize); // Allocate memory
    // Get Version resource

    GetFileVersionInfo(PChar(ParamStr(0)), 0, VerSize, PBlock);

    GetMem(pBuffer, 256); // Get predefined string
    sSubBlock := '\\StringFileInfo\\0409950\\CompanyName';
    VerQueryValue(pBlock, PChar(sSubBlock), pBuffer, Size);

    ShowMessage('Company Name: ' + StrPas(pBuffer));
    sSubBLock := '\\StringFileInfo\\0409950\\FileDescription';
    VerQueryValue(pBlock, PChar(sSubBlock), pBuffer, Size);

    ShowMessage('File Description: ' + StrPas(pBuffer));
    sSubBLock := '\\StringFileInfo\\0409950\\FileVersion';
    VerQueryValue(pBlock, PChar(sSubBlock), pBuffer, Size);

    ShowMessage('File Version: ' + StrPas(pBuffer));
    sSubBLock := '\\StringFileInfo\\0409950\\ProductName';
    VerQueryValue(pBlock, PChar(sSubBlock), pBuffer, Size);

    ShowMessage('Product Name: ' + StrPas(pBuffer));
    FreeMem(pBuffer);

  end;
end;
問題1-17: 如何產生resource DLL檔
在Delphi主選單, 選New --> DLL, Delphi便自動產生一個library專案。再將之前產生的sample.res檔連結至專案後存檔, 檔名為ResTest.dpr. 編譯後產生一名為ResTest.DLL。專案內容如下:

library ResTest;

{ Important note about DLL memory management: ShareMem must be the

  first unit in your library's USES clause AND your project's (select

  Project-View Source) USES clause if your DLL exports any procedures or

  functions that pass strings as parameters or function results. This

  applies to all strings passed to and from your DLL--even those that

  are nested in records and classes. ShareMem is the interface unit to

  the BORLNDMM.DLL shared memory manager, which must be deployed along

  with your DLL. To avoid using BORLNDMM.DLL, pass string information

  using PChar or ShortString parameters. }

uses

  SysUtils,

  Classes;

{$R *.RES}

{$R sample.RES} // 加入這一行, 將sample.res連結至專案

begin

end.
問題1-18: 如何使DLL或其它EXE內的resource
以ResTest.DLL檔為例, 將ResTest.DLL檔內的ResName為car的Icon變為Application的Icon

procedure TForm1.Button1Click(Sender: TObject);

var

  hLib: THandle; 

  hIcon: THandle;

begin

  hLib := LoadLibrary('C:\Test\RESTEST.DLL');

  if h <= 32 then

    ShowMessage(SysErrorMessage(GetLastError))

  else 
begin

    hIcon := LoadIcon(hLib, 'car');
    Application.Icon.Handle := hIcon;
    FreeLibrary(hLib);

  end;

end;
另外, 我們可以利用ResHacker這個軟體來檢視其它DLL或EXE內有哪些resource, 再比照上述方式來使用.

